
ABSTRACT  

The Gaussian quadrature algorithm approximates the likelihood of a generalized linear model with 
multiple nested levels using the Gauss-Hermite quadrature method of integration. While the 
approximations are close to exact, the calculation of these approximations in SAS are 
computationally demanding both in memory and time. To address these intense requirements, the 
SAS/STAT 14.1 update added the multilevel adaptive Gaussian quadrature algorithm of Pinheiro 
and Chao (2006) to PROC GLIMMIX using the METHOD = FASTQUAD option. Pinheiro and Chao’s 
algorithm reduces the number of integrations required while SAS processes the estimation of the 
conditional log-likelihoods required for this modelling technique. The reduction of integrations 
reduces computation memory and time.  Using a public dataset, this paper will examine the process 
of creating a multilevel model using adaptive Gaussian quadrature while comparing the run times of 
single level and multilevel adaptive Gaussian quadrature. 

INTRODUCTION  

Adaptive Gaussian Quadrature algorithms are based on the Gauss-Hermite Quadrature rule 
defined in the Handbook of Mathematical Functions of Abramowitz and Stegun (1964). The 
single level Adaptive Gaussian Quadrature algorithm of Pinheiro and Bates (1995) and 
multilevel Adaptive Gaussian Quadrature algorithm of Pinheiro and Chao (2006) both use 
importance sampling to approximate the log-likelihoods in Generalized Linear Mixed Models 
(GLMMs). Using the approximations of the log-likelihoods in GLMMs the maximum likelihood 
estimator for the parameters in the model may be derived. Multilevel GLMMs, such as 
students being nested within teachers, require exponentially more integrations as the 
number of levels increases. The increase in integrations can quickly cause the statistical 
software and the computers processing programs to quickly run out of computational 
memory. 

With the release of SAS/STAT® 14.1 the GLIMMIX procedure may now use the multilevel 
Adaptive Gaussian Quadrature algorithm. To use the multilevel algorithm instead of the 
single level algorithm, the FASTQUAD sub-option must be specified alongside 
METHOD=QUAD. With the FASTQUAD sub-option SAS is now able to process multilevel 
GLMMs more efficiently through integration reduction. The reduction in number of 
integrations leads to faster processing speeds and lower memory requirements.  

The largest advantage to the addition of the FASTQUAD sub-option is reduced computation 
memory and time. To adduce this advantage, we compared computation memory and run 

time between single level and multilevel Adaptive Gaussian Quadrature using a created 
dataset based on a SAS Support simulation (SAS Support, 2017). Due to a lack of memory 
the application of single level Adaptive Gaussian Quadrature to a multilevel model was 
unable to process. In place of the single level model, the number of quadrature points used 
in multilevel Adaptive Gaussian Quadrature were compared at levels of 3, 5, and 7. The 
increase in quadrature points leads to an increase in integrations required when 
approximating the log-likelihoods. This paper will discuss the theoretical differences 

between single level and multilevel Adaptive Gaussian Quadrature, the implementation of 
the FASTQUAD sub-option in the GLIMMIX procedure, and compare computational memory 
and run time requirements for using 3, 5 and 7 quadrature points for approximation. 

ADAPTIVE GAUSSIAN QUADRATURE 

WHAT IS ADAPTIVE GAUSSIAN QUADRATURE 

To understand the differences between single level and multilevel Adaptive Gaussian 
Quadrature we must start at its roots. Let 𝑝(𝑥) be a probability density function (pdf) and 
𝑓(𝑥) be a function which the pdf is integrated against. Gauss-Hermite quadrature is 

appropriate when the density of 𝑝(𝑥) has kernel 𝑒−𝑥2
 and the limits of integration span all 

real numbers. From formula 25.4.46 of Abramowitz and Stegun (1964), the Gauss-Hermite 
quadrature rule from which Adaptive Gaussian Quadrature is based follows the form 

∫ 𝑓(𝑥)𝑝(𝑥)𝑑𝑥 
∞

−∞

= ∫ 𝑒−𝑥2
𝑓(𝑥)𝑑𝑥 

∞

−∞

≈  ∑ 𝑤𝑖𝑓(𝑥𝑖)

𝑁

𝑖=1

 

Where N represents the number of quadrature points, 𝑤𝑖 and 𝑥𝑖 represent the weights and 
abscissas. These weights and abscissas are based on areas of high densities in the kernel 

function. Weights and densities for common kernels, such as the normal distribution, are 
calculated following the methods described in Hammer, P. C., & Wymore, A. W. (1957). 

In single level Generalized Linear Mixed Models (GLMMs) the marginal distribution of the 
random effects follows a ~𝑁(0, 𝜎2) distribution. Using the structure of the integrand in 
GLMMs, Adaptive Gaussian Quadrature transforms the calculation of multiple Gaussian 
quadrature rules into repetitive applications of one-dimensional Gaussian quadrature rules. 
In this case, the Gaussian quadrature rule is a deterministic version of a Monte Carlo 
integration algorithm in which random samples of the random effects, 𝑏𝑖, are generated 

from the ~𝑁(0, 𝜎2) distribution; and the samples and weights are fixed beforehand instead of 
randomly chosen. (Pinheiro and Chao 2006). To obtain increased efficiency, an importance 
sampling version of Gaussian quadrature is utilized. This importance sampling version, 
defined by Pinheiro and Bates (1995), is known as Adaptive Gaussian Quadrature. 

SINGLE LEVEL ADAPTIVE GAUSSIAN QUADRATURE 

In single level Adaptive Gaussian Quadrature for GLMMs the likelihood estimation, for a 
given group 𝑖, relies on the marginal density of the response vector 𝑦𝑖. The marginal density 

of 𝑦𝑖 is obtained by integrating the joint likelihood function of (𝑦𝑖 , 𝑏𝑖) with respect to the 
random effects vector 𝑏𝑖. For the purposes of this paper, let 𝛽 represent the vector of fixed 
effects, 𝑋𝑖 is the fixed effects regression matrix, 𝑍𝑖 is the random effects matrix, and ℎ = 𝑔−1 

is the inverse link function. The joint density of (𝑦𝑖 , 𝑏𝑖) when assuming an exponential link 
function in canonical form for appropriate functions 𝑎(. ), 𝑐(. ), and 𝑑(. ) is 

𝑝(𝑦𝑖 , 𝑏𝑖) =  
𝑒

[
𝑦𝑖

′(𝑋𝑖𝛽+𝑍𝑖𝑏𝑖)−𝑑(𝑋𝑖𝛽+𝑍𝑖𝑏𝑖)′1
𝑎(∅)

+𝑐(𝑦𝑖,∅)′−
𝑏𝑖

′(𝜎2)−1𝑏𝑖
2

]

[(2𝜋)
𝑞
2 |(𝜎2)|

1
2]

 

where 𝑞 is equal to the number of random effects. 

As previously stated, the marginal density of 𝑦𝑖, which the likelihood estimation relies upon, 
is obtained by integrating the joint likelihood function of (𝑦𝑖 , 𝑏𝑖) with respect to the random 

effects vector 𝑏𝑖. The marginal GLMM likelihood in the single level case can be written as 

𝑝(𝑦𝑖) =  ∫ 𝑝(𝑦𝑖 , 𝑏𝑖)𝑑𝑏𝑖 = (2𝜋)−
𝑞
2 |(𝜎2)|−

1
2 ∫ 𝑒𝑔(𝛽,(𝜎2,∅,𝑦𝑖,𝑏𝑖)𝑑𝑏𝑖 

and defining 𝑔(𝛽, 𝜎2, ∅, 𝑦𝑖 , 𝑏𝑖) as 

𝑔(𝛽, 𝜎2, ∅, 𝑦𝑖 , 𝑏𝑖) =
𝑦𝑖

′(𝑋𝑖𝛽 + 𝑍𝑖𝑏𝑖) − 𝑑(𝑋𝑖𝛽 + 𝑍𝑖𝑏𝑖)
′1

𝑎(∅)
+ 𝑐(𝑦𝑖 , ∅)′ −

𝑏𝑖
′(𝜎2)−1𝑏𝑖

2
 

Through a second-order Taylor expansion 𝑔(𝛽, 𝜎2, ∅, 𝑦𝑖 , 𝑏𝑖) is approximated around the value 
of 𝑏𝑖 which maximizes 𝑔(. , 𝑏𝑖). The second-order derivative of 𝑔(. , 𝑏𝑖) is negative definite 

resulting in 𝑔(. , 𝑏𝑖) being a strictly concave function of 𝑏𝑖. This strictly concave function 

allows the estimation of a unique point of maximum, �̂�𝑖, corresponding to the value of the 
second-order derivative of 𝑔(. , 𝑏𝑖) is equal to zero. The calculation of convergence on 𝑏𝑖 can 
be transferred to a least squares problem from which 𝑅𝑖 is defined as part of the Q-R 
decomposition in Thisted (1988). 

We can now approximate the importance distribution the Adaptive Gaussian Quadrature rule 

utilizes. In a single level GLMM the integrand is proportional to 𝑒𝑔(𝛽,𝜎2,∅,𝑦𝑖,𝑏𝑖) which is 

approximated by a ~𝑁(�̂�𝑖 , 𝑅𝑖
−1𝑅𝑖

−𝑇) density. The use of this importance distribution centers the 

grid of abscissas in the 𝑏𝑖 scale around the conditional nodes �̂�𝑖 and uses 𝑅𝑖 for scaling. To 
put everything together, let 𝑧𝑖, 𝑤𝑗 for 𝑗 = 1, … , 𝑁𝐺𝑄 denote the abscissas and weights for the 

one-dimensional Gaussian Quadrature rule with 𝑁𝐺𝑄 points based on the ~𝑁(0,1) kernel. 

Define the following: 

• 𝑧𝑗 = (𝑧𝑗1, … , 𝑧𝑗𝑞
)𝑇 

• �̃�𝑖𝑗 = �̂�𝑖 + 𝑅𝑖
−1𝑧𝑗 

• 𝑊𝑗 = 𝑒‖𝑧𝑗‖
2

∏ 𝑤𝑗𝑘

𝑞
𝑘=1  

The single level Adaptive Gaussian Quadrature rule is then 

∫ 𝑒𝑔(𝛽,𝜎2,∅,𝑦𝑖,�̂�𝑖+𝑅𝑖
−1𝑧)+

‖𝑧‖2

2 𝑒−
‖𝑧‖2

2 𝑑𝑧 ≅ (2𝜋)
𝑞
2|𝑅𝑖|

−1 ∑ …

𝑁𝐺𝑄

𝑗1=1

∑ [𝑒𝑔(𝛽,𝜎2,∅,𝑦𝑖,�̃�𝑖𝑗)𝑊𝑗]

𝑁𝐺𝑄

𝑗𝑞=1

 

and the single level GLMM Adaptive Gaussian Quadrature approximation of the log 

likelihood function is 

𝑙𝐴𝐺𝑄(𝛽, 𝜎2, ∅|𝑦) = 𝑀 log |∆| + ∑ {− log|𝑅𝑖| + log [∑ 𝑒[𝑔(𝛽,𝜎2,∅,𝑦𝑖,�̃�𝑖𝑗)]𝑊𝑗

𝑁𝐺𝑄

𝑗

]}

𝑀

𝑖=1

 

Where M is the number of groups and ∆ is a precision matrix from the least squares 

transformation. [3] 

MULTILEVEL ADAPTIVE GAUSSIAN QUADRATURE 

Multilevel Adaptive Gaussian Quadrature for GLMMs allows computational feasibility of 
approximating the log-likelihood function from data with a nested model structure. 
Extending the structure of multilevel GLMMs to the Adaptive Gaussian Quadrature, the 
approximation of the log-likelihood for the 𝑖𝑡ℎ group in a two-level GLMM becomes 

 
𝑙𝐴𝐺𝑄2(𝛽, 𝜎1

2, 𝜎2
2, ∅|𝑦𝑖)

= log|∆1| − log|𝑅11(𝑖)| + 𝑀𝑖 log|∆2| − ∑ log|𝑅22(𝑖𝑗)|
𝑀𝑖

𝑗=1

+ log ∑ {[𝑒𝑔21(𝛽,𝜎1
2,∅,𝑦𝑖,�̃�𝑖

(1)
(𝑘)𝑊𝑘

(1)
] × ∏ ∑ [𝑒

𝑔22(𝛽,𝜎2
2,∅,𝑦𝑖𝑗,�̃�𝑖

(1)(𝑘),�̃�𝑖𝑗
(2)

(𝑙,𝑘)
𝑊𝑙

(2)]

𝑁𝐺𝑄

𝑙

𝑀𝑖

𝑗=1

}

𝑁𝐺𝑄

𝑘=1

 

Where the full data approximation is  

𝑙𝐴𝐺𝑄2(𝛽, 𝜎1
2, 𝜎2

2, ∅|𝑦) = ∑ 𝑙𝐴𝐺𝑄2(𝛽, 𝜎1
2, 𝜎2

2, ∅|𝑦𝑖)
𝑀𝑖

𝑖=1
 

The approximation requires 𝑁𝐺𝑄
𝑞1 evaluations of 𝑔21(. ) and 𝑀𝑖𝑁𝐺𝑄

𝑞1+𝑞2 evaluations of 𝑔22(. ) which 

increases linearly with the number of level two groups, 𝑀𝑖. If single level Adaptive Gaussian 
Quadrature were to be directly applied to a two-level model the approximation would 

require 𝑁𝐺𝑄
𝑞1+𝑀𝑖𝑞2 evaluations of the 𝑔(. ) function. The increase in computational speed and 

memory efficiency comes from using 𝑀𝑖𝑁𝐺𝑄
𝑞1+𝑞2 evaluations instead of 𝑁𝐺𝑄

𝑞1+𝑀𝑖𝑞2 evaluations of 

the respective 𝑔(. ) functions. This difference in evaluations is a product of theoretical 
change in the Q-R decompositions for the level 2 arrays in the data. The multilevel Adaptive 
Gaussian Quadrature algorithm reduces the number of conditional log-likelihood evaluations 

by inverting the least squares upper-triangular matrices; and the log-determinants used in 
the computations become the sum of the logarithms of the absolute diagonal values of the 
corresponding upper-triangular matrices. Pinheiro and Chao (2006) documents the 
application of inverting the upper-triangular matrices to the Adaptive Gaussian Quadrature 
rule in detail. [4] 

PROC GLIMMIX 

The dataset used in the following section was simulated using SAS code which can be found 
in the appendix. The simulation creates a multilevel dataset with 20 schools, 30 teachers at 
each school, and between 16 and 30 students for each teacher. Each student has a value 
for the fixed effect x1. The response variable in the model follows a binomial response 
tracking whether a student had the flu. The following code applies a direct application of the 
single level Adaptive Gaussian Quadrature approximation to the multilevel simulated dataset 

using 3 quadrature points: 

   proc glimmix data=sim method=quad(qpoints=3); 

 class school teacher; 

 model Flu = x1 /dist=binomial link=logit; 

 random int / subject=school; 

 random int / subject=teacher(school); 

   run; 

Unless the dataset used is small, SAS will return the following error message: 

ERROR: Insufficient resources to perform adaptive quadrature with 3 

quadrature points. METHOD=LAPLACE, corresponding to a single point, might      

provide a computationally less intensive possibility. 

The reason this error message displays is because the number of integrations the single 

level Adaptive Gaussian Quadrature approximation requires SAS to perform. In the context 

of the simulated data, this procedure asks SAS to perform 𝑁𝐺𝑄
𝑞1+𝑀𝑖𝑞2 = (3)1+20∗1 = 10,460,353,203 

evaluations of the conditional log-likelihoods. 

To use the multilevel Adaptive Gaussian Quadrature approximation the METHOD=QUAD 
option with the FASTQUAD sub-option must be used. The only difference in the code is the 
addition of the FASTQUAD sub-option shown below: 

   proc glimmix data=sim method=quad(fastquad qpoints=3); 

 class school teacher; 

 model Flu = x1 /dist=binomial link=logit; 

 random int / subject=school; 

 random int / subject=teacher(school); 

   run; 

This time, we receive no error message because using the FASTQUAD sub-option asks SAS 

to perform 𝑁𝐺𝑄
𝑞1+𝑞2 = (3)1+1 = 9 evaluations of the conditional likelihoods. The output for the 

GLIMMIX procedure can be found in the appendix. 

MEMORY AND RUNTIME IMPROVEMENT 

Since the single level Adaptive Gaussian Quadrature approximation may not be successfully 
run when applied to a multilevel model, the memory usage and runtimes of multilevel 
models with quadrature points of 3, 5 and 7 will be compared. The computer which 
processed each GLIMMIX procedure is a Lenovo™ Thinkpad® L480. The L480 uses an 
Inter® Core™ i5-8350U CPU with an internal clock of 1.70Ghz, 8.00 GB of RAM, and a 64-
bit Windows® 10 Operating System. SAS Enterprise Guide was used to run the SAS 
programs. 

There was a total of 10 trials created following the format of the simulated dataset 

mentioned in the previous section. For each trial a GLIMMIX procedure with quadrature 
points of 3, 5, and 7 were ran one at a time. The runtime and memory usage statistics were 
recorded from the log after running the code: 

   options fullstimer; 

 

After which the log information will now display: 
NOTE: Convergence criterion (GCONV=1E-8) satisfied. 
NOTE: At least one element of the gradient is greater than 1e-3. 
NOTE: PROCEDURE GLIMMIX used (Total process time): 
      real time           12.42 seconds 

      user cpu time       11.98 seconds 

      system cpu time     0.42 seconds 

      memory              5897.03k 

      OS Memory           23640.00k 

      Timestamp           02/16/2020 07:27:51 PM 

 

From which the real time and the memory was recorded for each of the procedures ran. A 

means procedure was used to summarize the results on the following page: 

Quadrature 
points 

Measure Mean Standard 
Deviation 

Minimum Median Maximum 

3 Runtime 
(s) 

12.35 0.90 11.07 12.30 13.74 

Memory 
(k) 

5619.55 8.33 5611.28 5618.05 5641.42 

5 Runtime 
(s) 

23.05 1.90 20.53 23.56 26.15 

Memory 
(k) 

5617.71 11.97 5585.34 5619.70 5630.19 

7 Runtime 
(s) 

39.61 2.74 35.92 40.30 43.53 

Memory 
(k) 

5620.87 4.25 5613.60 5619.67 5627.67 

Table 1. MEANS Procedure: Runtime and Memory Statistics for FASTQUAD 

 

The mean runtime of the GLIMMIX procedure in seconds increases with the number of 
quadrature points used. This can be explained by the increase in the number evaluations of 
the conditional log-likelihoods: 

• 3 quadrature points: 𝑁𝐺𝑄
𝑞1+𝑞2 = (3)1+1 = 9 evaluations of the conditional log-likelihoods. 

• 5 quadrature points: 𝑁𝐺𝑄
𝑞1+𝑞2 = (5)1+1 = 25 evaluations of the conditional log-likelihoods. 

• 7 quadrature points: 𝑁𝐺𝑄
𝑞1+𝑞2 = (7)1+1 = 49 evaluations of the conditional log-likelihoods. 

As expected, when SAS is asked to perform more evaluations of the conditional log-
likelihoods the program takes longer to run.  

The standard deviations for the memory requirements increased from 3 points to 5 but 
decreased from 5 to 7. While this result was not expected, it also is not surprising. This is 
due to the limitation of running multiple procedures using the same dataset. After a couple 
procedures are completed, results may be stored in either the operating system, within 
SAS, or another process on the computer. The caching of these results likely lead to 
subsequent procedures having lower variances. 

CONCLUSION 

The addition of the FASTQUAD sub-option in the GLIMMIX procedure for SAS/STAT 14.1 
allows the use of the multilevel Adaptive Gaussian Quadrature algorithm of Pinheiro and 
Chao (2006). The multilevel algorithm reduces the number of evaluations of the conditional 
log-likelihoods by inverting the least squares upper-triangular matrices. By decreasing the 
number of integrations of the conditional log-likelihoods SAS can process multilevel models 
faster and more efficiently. The reduction in integrations allows users to use additional 
quadrature points to increase accuracy of the evaluations performed to derive the model’s 
parameter estimates. 

APPENDIX 

CODE 

 
/* Create the simulation dataset.*/ 

data sim; 

 do trial = 1 to 10; /* Create 10 trials w/ random seeds*/ 

 call streaminit(rand('uniform', 10000, 10000000)); 

 do school = 1 to 20; /* Create 20 schools */ 

  rs = rand('normal')*2; 

  do teacher = 1 to 30; /*Create 20 teachers*/ 

   rt = rand('normal'); 

   do student = 1 to ceil(rand('uniform')*14)+15; 

    x1 = rand('normal'); 

    /*Create linp relating the random effects*/ 

    linp = -3 + x1 + rs + rt; 

    pi = 1/(1+exp(-linp)); 

    /* Create a Flu Response based on linp*/ 

    Flu = ranbin(0,1,pi); 

    output; 

   end; 

  end; 

 end; 

 end; 

run; 

 

 

*Simulation code; 

 

/* Creating a call function to quickly change trial number*/ 

%let trialnum = 1; 

 

/* Telling SAS to print the full usage statistics in the log*/ 

options fullstimer; 

 

/*Single level Adaptive Gaussian Quadrature applied to multilevel model*/ 

proc glimmix data=sim method=quad; 

 where trial=&trialnum; 

 class school teacher; 

 model Flu = x1 /dist=binomial link=logit; 

 random int / subject=school; 

 random int / subject=teacher(school); 

run; 

 

/*Multilevel model with 3 quadrature points*/ 

proc glimmix data=sim method=quad(fastquad qpoints=3); 

 where trial=&trialnum; 

 class school teacher; 

 model Flu = x1 /dist=binomial link=logit; 

 random int / subject=school; 

 random int / subject=teacher(school); 

run; 

 

/*Multilevel model with 5 quadrature points*/ 

proc glimmix data=sim method=quad(fastquad qpoints=5); 

 where trial=&trialnum; 

 class school teacher; 

 model Flu = x1 /dist=binomial link=logit; 

 random int / subject=school; 

 random int / subject=teacher(school); 

run; 

 

/*Multilevel model with 7 quadrature points*/ 

proc glimmix data=sim method=quad(fastquad qpoints=7); 

 where trial=&trialnum; 

 class school teacher; 

 model Flu = x1 /dist=binomial link=logit; 

 random int / subject=school; 

 random int / subject=teacher(school); 

run; 

 

/*Importing dataset with the 10 trial’s runtime and memory usage*/ 

proc import datafile=’…\SimulationData.xlsx' replace out=simres; 

run; 

 

/*Creating a dataset, sorting and printing the means with the statistics 

desired*/ 

data results; 

 set simres(drop=Trial); 

run; 

proc sort data=results; 

 by Qpoints; 

run; 

proc means data=results nonobs min mean std median max maxdec=2; 

 by Qpoints; 

 var Runtime 'Memory(k)'n; 

run; 

OUTPUT 

When the GLIMMIX procedure completes, the following output is given by default: 

 
Output 1. Model Information 

 

 
Output 2. Class Level Information 

 

 
Output 3. Observations Read/Used 

Output 4. Matrix Dimensions 

 

 
Output 5. Optimization Information 

 

 
 

 
Output 6. Iteration History 

 

 
Output 7. Model Fit Statistics for the Multilevel Model 

Output 8. Model Fit Statistics for the Conditional Distribution 

 

 
Output 9. Covariance Parameter Estimates 

 

 
Output 10. Type III Tests of Fixed Effects 

  


